Math 174
Calculus II

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 17

The area of a Surface of Revolution

Initializations

```
> restart;
  with(plots):
```

17.1 Area of a Surface of Revolution

If the curve $y = f(x)$, $a \le x \le b$ is rotated about the x axis, then the area of the resulting surface is given by

$$2\pi \int_a^b f(x) \sqrt{1 + (f'(x))^2}\, dx$$

The mathematical details will be provided in class.

Examples

Example 17.1.1

Find the area of the surface of revolution obtained by revolving the curve

$$y = \sin x,\ 0 \le x \le \frac{3}{4}\pi$$

about the x axis.

Solution
Use the formula provided above.

```
> f:=x->sin(x);
```

$$f := x \rightarrow \sin(x) \qquad \textbf{(2.1.1.1)}$$

```
> e1:=2*Pi*Int(f(x)*sqrt(1+D(f)(x)^2), x=0..3*Pi/4);
```

$$e1 := 2\pi \left(\int_0^{\frac{3}{4}\pi} \sin(x) \sqrt{1 + \cos(x)^2}\, dx \right) \qquad \textbf{(2.1.1.2)}$$

```
> surface_area:=simplify(value(e1));
  evalf(surface_area);
```

$$surface\_area := \frac{1}{2}\pi \left(2\sqrt{2} + 2\ln\left(1+\sqrt{2}\right) + \sqrt{3} - 2\ln(2) + 2\ln\left(\sqrt{2} + \sqrt{2}\sqrt{3}\right)\right)$$

$$12.00117140 \qquad \textbf{(2.1.1.3)}$$

It is possible to visualize the resulting solid of revolution. The plotting routine makes use of a parametrization of the surface. You will learn the details in Calculus III. At this time you are not required to be able to do this.

```
> plot3d([x, sin(x)*cos(u), sin(x)*sin(u)], x=0..3*Pi/4, u=
  0..2*Pi, style=patch, orientation=[-60, 70], axes=boxed,
  labels=[x, y, z], scaling=constrained);
```




```
>
```

Example 17.1.2

Rotate the curve

$$x = \frac{\left(y^2+2\right)^{\frac{3}{2}}}{3}, 1 \le y \le 2$$

about the *x* axis and compute the area of the resulting surface of revolution.

Solution

This is an example where integration over y is appropriate. If dS denotes the surface area of a slice of the surface in a plane perpendicular to the x axis, then

$$dS = 2\pi\, r\, ds$$

where r denotes the radius of the slice and ds denotes the length of the piece of arc that generates the slice.

Therefore, dS can be written as

$$dS = 2\pi f(x)\sqrt{1+\left(\frac{dy}{dx}\right)^2}\,dx$$

but also as

$$dS = 2\pi y\sqrt{1+\left(\frac{dx}{dy}\right)^2}\,dy$$

We now make use of the second representation.

```
> g:=y->1/3*(y^2+2)^(3/2);
```

$$g := y \rightarrow \frac{1}{3}\left(y^2+2\right)^{3/2} \qquad \textbf{(2.1.2.1)}$$

```
> e1:=2*Pi*Int(y*sqrt(1+D(g)(y)^2), y=1..2);
```

$$e1 := 2\,\pi\left(\int_1^2 y\sqrt{\left(1+y^2\right)^2}\,dy\right) \qquad \textbf{(2.1.2.2)}$$

```
> e2:=simplify(e1);
```

$$e2 := 2\,\pi\left(\int_1^2 y\left(1+y^2\right)dy\right) \qquad \textbf{(2.1.2.3)}$$

```
> surface_area:=value(e2);
  evalf(surface_area);
```

$$surface\_area := \frac{21}{2}\,\pi$$

$$32.98672287 \qquad \textbf{(2.1.2.4)}$$

```
>
```